# Membrane Worlds

## Preamble

Author: Mikael Beuthe
Purpose: computation of tidal/load Love numbers of ‘membrane worlds’, or icy satellites having a subsurface ocean under a thin crust
Version: 1.0
Date: 16 June 2015

The notebook is based on the following papers:
MP (for ‘Membrane Paradigm’) = Beuthe (2015a), Tides on Europa: The membrane paradigm, Icarus 248, 109-134; arXiv.org:1410.4735v1
MW (for ‘Membrane Worlds’) = Beuthe (2015b), Tidal Love numbers of membrane worlds: Europa, Titan and Co., Icarus (in press, available online) doi:10.1016/j.icarus.2015.06.008; arXiv.org:1504.04574v2 (revised version)
If you use this notebook to obtain results for a publication, or include part of it in your own code, please cite the papers above.



## Table of contents

## Part III: Other examples

### 0. Constants

### 1. Effective viscoelastic parameters



### 2. $l_2 - h_2$ relation



### 3. $k_2 - h_2$ relation



### 4. Static Love numbers



### 5. Dynamical Love numbers



### 6. A simple thick shell model



# Part I: Pedestrian example

Let’s say you want to check the effect of crustal rheology on Europa’s tidal deformation and dissipation. For this, you need to compute Europa’s tidal Love numbers.

As an application, we will evaluate the heat dissipated in the whole satellite and determine the specific contributions of the crust and mantle.

# 1. Love numbers

## ▪ First thing to do

Evaluate the section II.2. Formulas.
You will be asked to do it anyway, because the formulas section is an Initialization Cell that should be evaluated before any other.
Also, clear the variables used in the example:

```
Clear[y, ξ0, μmHat]
Clear[ω, ecc, R, Rm, ρb, g, ρbar, ρ, δρ, d, μEm, μE, νE]
Clear[ϵ, δm, η, μ, ν, χ, f, δ1, δ2]
Clear[obli, Gconstant]
```

## ▪ Outline of the method

The membrane formulas for the gravitational Love number of degree n is

```
kn[μhat, ν, χ, δρhat, ϵ, ξ0, hn0, n]
```

The formula for the radial Love number of degree n is similar:

```
hn[μhat, ν, χ, δρhat, ϵ, ξ0, hn0, n]
```

The parameters ($\mu$hat, $\nu$, $\chi$, $\delta\rho$hat, $\epsilon$) are relevant to the shallow interior (crust and crust-ocean boundary).
The parameters ($\xi$0, hn0) are relevant to the deep interior (ocean and below).

In the membrane approach, computing Love numbers is a two-step procedure:
1) evaluate the fluid-crust Love number *hn0*, i.e. the radial Love number of a model in which the crust is fluid and may differ in density from the physical crust.
2) add the crust modeled as a membrane, which means that its viscoelasticity has been integrated over crust thickness.
Any arbitrarily complicated depth-dependent rheology is then summed up into three complex numbers: the effective viscoelastic parameters.

## ▪ Physical model: general structure

For Europa, we know that the surface radius is R=1560.8 km, the bulk density is $\rho$b=3013 kg/m3, and the surface gravity is g=1.315 m/s2.
The bottom of the ocean must be be at a depth of less than 170 km; assume here that this bound corresponds to the radius of the mantle.
The density of the crust ($\rho$bar) and ocean ($\rho$) are unknown: assume that $\rho$bar=950 kg/m3 and $\rho$=1050 kg/m3.
The crust-ocean density contrast is defined as $\delta\rho$= $\rho$bar-$\rho$=-100kg/m3. For the crust thickness, assume that d=25 km (no attempt is made here to fit the observed moment of inertia).

For later use, note that the tidal angular frequency is $\omega$=2.048 x $10^{-5}$ rad/s and that the orbital eccentricity is ecc=0.0094.
These values define the following replacement rule:

```
rulePedestrian =
  {ω -> 2.048 * 10^-5, ecc → 0.0094, R -> 1560.8 * 10^3, Rm -> 1560.8 * 10^3 - 170. * 10^3,
   ρb → 3013., g → 1.315, ρbar → 950., ρ → 1050., δρ → -100., d → 25. * 10^3}
```

$\{\omega \to 0.00002048,\ ecc \to 0.0094,\ R \to 1.5608\times 10^6,\ Rm \to 1.3908\times 10^6,$
$\rho b \to 3013.,\ g \to 1.315,\ \rho bar \to 950.,\ \rho \to 1050.,\ \delta\rho \to -100.,\ d \to 25000.\}$

## ▪ Choose a fluid-crust model

The first step consists in choosing a model for the deep interior. The subsurface ocean partially decouples the mantle from the crust, so that the core and mantle mainly affect the Love numbers

through their bulk mass. For this reason, we consider here a very simple deep interior model: core and mantle are treated as one incompressible viscoelastic layer, below a homogeneous and incompressible ocean (density stratification of the ocean is not so important for Europa because the ocean is not very deep). However, nothing forbids you to use a more complicated interior model, with any number of layers or even a continuous radial variation of the physical parameters.
Our choice is thus the VL model, where 'VL' stands for 'Viscoelastic/Liquid': this is a two-layer incompressible model made of viscoelastic mantle and a surface ocean. This choice implies that the fluid-crust density is equal to the ocean density.
The degree-two radial Love number of the VL model is given by:

```
hVL0[y, ξ0, μmHat, 2]
```

$$\left(5\,y^4\,\mu\text{mHat} + \frac{5}{19}\,(1-\xi 0)\,\left(\left(2+3\,y^5\right)\,(1-\xi 0) + 5\,y^3\,\xi 0\right)\right) \Big/ \left(y^4\,\mu\text{mHat}\,(5-3\,\xi 0) + \frac{1}{19}\,(1-\xi 0)\,\left(-9\,y^5\,(1-\xi 0)\,\xi 0 + (5-3\,\xi 0)\,\left(2\,(1-\xi 0) + 5\,y^3\,\xi 0\right)\right)\right)$$

In this formula, the variables are:
*y=Rm/R*: relative radius of the mantle with respect to the surface,
$\xi 0 = \rho/\rho b0$ : ocean-to-bulk density ratio,
$\mu$mHat = $\mu$m/($\rho$b0*g0*R) : nondimensional shear modulus of the mantle.

## Fluid-crust model: density and gravity

Now, the bulk density $\rho$b0 and the surface gravity g0 of the fluid-crust model are *not* the same as those of the physical model.
This is because the fluid crust has the density of the ocean, instead of having the density of the physical crust.
This changes the bulk density and surface gravity as follows:

```
ρb0[ρb, δρ, ϵ]
g0[g, ρb, δρ, ϵ]
```

$$-3\,\delta\rho\,\epsilon + \rho b$$

$$g\left(1 - \frac{3\,\delta\rho\,\epsilon}{\rho b}\right)$$

where $\epsilon$ is the relative crust thickness: $\epsilon$=d/R. These relations are correct up to first order in $\epsilon$=d/R.
For the parameters chosen above, the bulk density and surface gravity of the fluid-crust model are

$$\left\{\rho b0\left[\rho b,\ \delta\rho,\ \frac{d}{R}\right],\ g0\left[g,\ \rho b,\ \delta\rho,\ \frac{d}{R}\right]\right\} \text{ /. rulePedestrian}$$

```
{3017.81, 1.3171}
```

## Fluid-crust model: mantle rheology

Before computing the fluid-crust Love numbers, we need to consider mantle rheology.
For the elastic shear modulus of the mantle, assume that $\mu$Em=40 GPa.
If the mantle is viscoelastic, its shear modulus is not 40 GPa but a complex number depending on the forcing frequency and on the rheological model.
You could choose any rheological model (Burgers, Andrade etc), but assume here for simplicity that the rheology is Maxwell:

```
μV[μE, δm]
```

$$\frac{\mu E}{1 - i\,\delta m}$$

where $\delta_m$ is a dimensionless parameter inversely proportional to tidal frequency $\omega$ and viscosity $\eta_m$: $\delta_m = \mu_E\,/\,(\omega * \eta_m)$.

If $\delta_m = 0$, the body is purely elastic;
if $\delta_m = 1$, the body is in the critical state;
if $\delta_m >> 1$, the body is fluid-like.

The forcing angular frequency is $\omega$=2.048 x $10^{-5}$ rad/s;

the viscosity of the core-mantle system could be $10^{21}$Pa.s, as the Earth's mantle. In that case, the $\delta_m$ parameter is

```
μEm/(ω * ηm) /. rulePedestrian /. {μEm -> 40. * 10^9, ηm → 1. * 10^21}
```

```
1.95312 × 10^-6
```

which is so small ($\delta$<<1) that the core-mantle system can be treated in practice as being purely elastic at tidal frequencies (i.e. you can set $\delta = 0$).
Just for the fun, let's assume that the viscosity of the core-mantle system is such that its rheology is not far from critical: $\delta = 0.1$:

```
ruleMantleRheology = {μEm -> 40. * 10^9, δm → 0.1};
```

## Fluid-crust Love numbers

The nondimensionless parameters are expressed in terms of physical parameters with the rule

```
ruleMantleParam =
  {y -> Rm/R, ξ0 -> ρ/ρb0[ρb, δρ, d/R], μmHat -> μV[μEm, δm]/(ρb0[ρb, δρ, d/R] * g0[g, ρb, δρ, d/R] * R)};
```

Now evaluate the radial fluid-crust Love number:

```
h2FluidCrust =
 hVL0[y, ξ0, μmHat, 2] /. ruleMantleParam /. ruleMantleRheology /. rulePedestrian
```

```
1.27839 - 0.00142737 i
```

In comparison with the elastic case, the real part of the fluid-crust Love number only changes at the 5th decimal, but the imaginary part (representing dissipation) is non-zero at the 3rd decimal. Note that mantle dissipation is still rather low for $\delta$=0.1: it increases until $\delta$=54 before decreasing again.

As the surface is hydrostatic, the gravitational fluid-crust Love number is related to the radial one by

```
k2FluidCrust = h2FluidCrust - 1
```

```
0.278388 - 0.00142737 i
```

## Adding the crust

The second step consists in adding the crust.
Assume that the crust is made of an elastic top layer (or lithosphere) and of a bottom layer that may be convecting.

The relative thickness of the top layer is $f = d_{\text{top}} / d$.

The rheology of each layer is Maxwell, the top and bottom layers being parameterized by $\delta_1$ and $\delta_2$, respectively.
Thus the effective viscoelastic parameters of the crust are given by the following replacement rule

```
ruleEffective =
  {μ -> μbar[μE, f, δ1, δ2], ν -> νbar[νE, f, δ1, δ2], χ -> χbar[νE, f, δ1, δ2]};
```

For example, the two-layer effective shear modulus is given by

```
μbar[μE, f, δ1, δ2]
```

$$\frac{f\,\mu E}{1 - i\,\delta 1} + \frac{(1 - f)\,\mu E}{1 - i\,\delta 2}$$

The elastic shear modulus of the icy crust is 3.5 GPa. The elastic Poisson's ratio is $\nu_E = 0.33$.
For the boundary between top and bottom layers, assume that *f = 0.3*.
Regarding the rheology, the top layer has $\delta_1$=0 (because it is elastic) and the bottom layer has $\delta_2$=1 (critical state).

```
ruleCrustRheology = {μE → 3.5 * 10^9, νE → 0.33, f → 0.3, δ1 → 0, δ2 → 1};
μbar[μE, f, δ1, δ2] /. ruleCrustRheology
χbar[νE, f, δ1, δ2] /. ruleCrustRheology
νbar[νE, f, δ1, δ2] /. ruleCrustRheology
```

$2.275 \times 10^9 + 1.225 \times 10^9\, i$

0.399293 + 0.163474 i

0.392317 − 0.0446284 i

## Physical Love numbers

It is now possible to compute the gravitational Love number $k_2$ with the formula kn[$\mu$hat, $\nu$, $\chi$, $\delta\rho$hat, $\epsilon$, $\xi$0, hn0, 2]

```
k2 = kn[μ/(ρ * g * R), ν, χ, δρ/ρ, d/R, ρ/ρb0[ρb, δρ, d/R], h2FluidCrust, 2] /. rulePedestrian /.
    ruleEffective /. ruleCrustRheology
```

0.263731 − 0.00698964 i

as well as the radial Love number $h_2$ with the formula hn[$\mu$hat, $\nu$, $\chi$, $\delta\rho$hat, $\epsilon$, $\xi$0, hn0, 2]

```
h2 = hn[μ/(ρ * g * R), ν, χ, δρ/ρ, d/R, ρ/ρb0[ρb, δρ, d/R], h2FluidCrust, 2] /. rulePedestrian /.
    ruleEffective /. ruleCrustRheology
```

1.21751 − 0.0262044 i

The tangential Love number $l_2$ is related to $h_2$ by the $l_2 - h_2$ relation:

```
l2 = lnhn[ν, 2] * h2 /. ruleEffective /. ruleCrustRheology
```

0.314266 − 0.0142416 i

The $l_2 - h_2$ relation is only valid at order $\epsilon^0$, i.e. in the limit of zero crust thickness, whereas the formulas above for $k_2$and $h_2$ are valid up to order $\epsilon$.
Thus the imaginary part of $l_2$ (which is of order $\epsilon$) cannot be fully trusted.

## Compact formulation

If you want to evaluate the above model for other sets of parameters, define a $k_2$ function that depends directly on the physical parameters:

```
k2VLVV[R_, d_, Rm_, ρb_, ρbar_, ρ_, g_, μE_, νE_, f_, δ1_, δ2_, μEm_, δm_] :=
  kn[μbar[μE, f, δ1, δ2]/(ρ * g * R), νbar[νE, f, δ1, δ2],
   χbar[νE, f, δ1, δ2], (ρbar - ρ)/ρ, d/R, ρ/ρb0[ρb, ρbar - ρ, d/R],
   hVL0[Rm/R, ρ/ρb0[ρb, ρbar - ρ, d/R], μV[μEm, δm]/(ρb0[ρb, ρbar - ρ, d/R] * g0[g, ρb, ρbar - ρ, d/R] * R), 2], 2]
```

In the function name, 'VLVV' denotes Viscoelastic/Liquid/Viscoelastic/Viscoelastic for the 4-layer structure: viscoelastic core-mantle, (liquid) ocean, and two viscoelastic crust layers.

Do the same for $h_2$:

```
h2VLVV[R_, d_, Rm_, ρb_, ρbar_, ρ_, g_, μE_, νE_, f_, δ1_, δ2_, μEm_, δm_] :=
 hn[μbar[μE, f, δ1, δ2]/(ρ * g * R), νbar[νE, f, δ1, δ2],
  χbar[νE, f, δ1, δ2], (ρbar - ρ)/ρ, d/R, ρ/ρb0[ρb, ρbar - ρ, d/R],
  hVL0[Rm/R, ρ/ρb0[ρb, ρbar - ρ, d/R], μV[μEm, δm]/(ρb0[ρb, ρbar - ρ, d/R] * g0[g, ρb, ρbar - ρ, d/R] * R), 2], 2]
```

As an example, plot $k_2$and $h_2$ in terms of crust thickness.
Assume that the crust top layer and the core-mantle system are purely elastic ($\delta_1 = 0$ and $\delta_m = 0$). Consider three rheology models for the crust bottom layer: $\delta_2 = 0$ (elastic, black), $\delta_2 = 1$ (critical, red), $\delta_2 = 100$ (fluid-like, blue).
Plot the real parts:







Plot the imaginary parts:




# 2. Dissipation

## ■ Total dissipation

As an application of the Love numbers computed above, we will now evaluate the heat dissipated in the whole satellite and determine the specific contributions of the crust and mantle. The satellite is synchronously rotating with spin rate $\omega$, orbital eccentricity *ecc* and obliquity *obli*.

The global heat flow due to tidal dissipation in the whole satellite is equal to the product of Im($k_2$) and of a global factor independent of the internal structure (see MP-99; MW-100):

```
Psi0 = (21/5 ecc^2 + 3/5 (Sin[obli])^2) /. {obli → 0} /. rulePedestrian
GlobalFac = -5/2 * (ω * R)^5 / Gconstant * Psi0 /. {Gconstant → 6.67 * 10^-11} /. rulePedestrian
```

```
0.000371112
```

```
-4.642 × 10^14
```

Here the obliquity has been set to zero.
Psi0 represents the nondimensional surface average of the squared norm of the tidal potential.

The global heat flow (in W) due to tidal dissipation in the whole satellite is

```
TotalDissip = GlobalFac * Im[k2] /. rulePedestrian
```

```
3.24459 × 10^12
```

## ■ Dissipation within the crust

The membrane approach also provides a formula for the dissipation within the crust (MP-98; MW-101), which is proportional to $|h_2|^2$ and to Im($\Lambda$), where $\Lambda$ is the membrane spring constant given by

```
Λ2 = Λ[μ / (ρ * g * R), ν, d/R, 2] /. rulePedestrian /. ruleEffective /. ruleCrustRheology
```

```
0.0353813 + 0.0179803 i
```

The global heat flow (in W) due to dissipation within the crust is given by (MP-98 and 138;

MW-101)

```
CrustDissip = GlobalFac * (-3/5) * ρ/ρb * (Abs[h2])^2 * Im[Λ2] /. rulePedestrian
```

$2.58814 \times 10^{12}$

## ■ Dissipation within the core and mantle

Finally, the membrane approach predicts that the global heat flow due to dissipation in the core and mantle is proportional to the imaginary part of the gravitational fluid-crust Love number. The exact relation involves the global factor given above and the squared norm of the scaling factor $\zeta$. The latter represents the relative reduction in radial displacement at the mantle-ocean boundary due to the presence of the crust (MW-95 and 96):

```
ζ = (1 - 3 * d/R * δρ/ρb) (k2 + 1)/(k2FluidCrust + 1) /. rulePedestrian
```

$0.990117 - 0.00437076\, i$

The global heat flow (in W) due to dissipation within the mantle is given by (MW-103):

```
MantleDissip = GlobalFac * (Abs[ζ])^2 * Im[k2FluidCrust] /. rulePedestrian
```

$6.49567 \times 10^{11}$

## ■ Heat budget

Now check that the crust and mantle contributions add up to the total heat flow:

```
CrustDissip + MantleDissip
TotalDissip
```

$3.23771 \times 10^{12}$

$3.24459 \times 10^{12}$

The heat budget is correct up to an error of 0.2%:

```
100 * (TotalDissip - (CrustDissip + MantleDissip))/TotalDissip
```

0.21223

As noted above, dissipation within the mantle is in reality much smaller because the mantle is nearly elastic at tidal frequency.

# Part II: Definitions

This part includes
1) a short description of the various symbols and
2) a section listing all the membrane formulas you could need to compute Love numbers

## 1. Notation

Beware:
- There is one important notation difference between the papers MP and MW.
  $\overline{\rho}$ denotes the bulk density in the former paper but the mean crust density in the latter.
  This notebook follows the latter convention.
- Equation numbers in MW refer to the published version or, equivalently, to the revised version of arXiv




A list of the parameters relevant to the shallow interior is given in MW-Table 4

**Size (dimensional)**
R: surface radius
Rm: mantle radius
d: crust thickness
dO: ocean thickness

**Size (nondimensional)**
ϵ=d/R: relative crust thickness (MW-28)
ϵO=dO/R: relative ocean thickness
fϵ=$d_{\text{top}} / d$ if the crust is made of two uniform layers of thicknesses $d_{\text{top}}$and $d_{\text{bot}}$
x=(R-d)/R: relative radius of the crust-ocean boundary with respect to the surface
y=Rm/R: relative radius of the mantle with respect to the surface
z=Rm/Rϵ: relative radius of the mantle with respect to the crust-ocean boundary (Rϵ=R-d)

**Density**
$\rho$b: bulk density
$\rho$: density of the top layer of the ocean
$\rho$bar: mean crust density (MW-31)
$\delta\rho$=$\rho$bar-$\rho$: crust-ocean density contrast (MW-32)
$\delta\rho$hat=$\delta\rho$/$\rho$: crust-ocean density contrast, nondimensionalized by the density of the top layer of the ocean.

**Gravity**
g: surface gravity (MW-34)

**Elasticity**
$\mu$Emantle: elastic shear modulus of the mantle
$\mu$E: elastic shear modulus of the icy crust
$\nu$E: elastic Poisson's ratio of the icy crust
$\chi$E: elastic compressibility factor of the icy crust (MW-12)
$\mu$Ehat = $\mu$E/($\rho$*g*R) : elastic shear modulus of the crust, $\mu$bar, nondimensionalized by $\rho$gR ($\rho$ is the ocean density)

**Viscoelasticity**
$\delta = \mu_E / (\omega * \eta)$ : dimensionless parameter inversely proportional to tidal frequency $\omega$ and viscosity $\eta$ (as in Wahr et al. 2009).

$\mu$ (or $\mu$bar if considered as a function): effective shear modulus of the icy crust (MW-42)
$\nu$ (or $\nu$bar if considered as a function): effective Poisson's ratio of the icy crust (MW-43)
$\chi$ (or $\chi$bar if considered as a function): effective compressibility factor of the icy crust (MW-38 and 45)
$\mu$hat = $\mu$bar/($\rho$*g*R) : the effective shear modulus of the crust, $\mu$bar, nondimensionalized by $\rho$gR (MW-46)
$\mu$mHat = $\mu$m/($\rho$b0*g0*R) : nondimensional shear modulus of the mantle

**Varia**
n: harmonic degree
q$\omega$ = $\omega^2 \, R / g$ : dynamical parameter

**Notation specific to fluid-crust models**
In general, attaching '0' to a variable means that we are dealing with a fluid-crust model instead of the physical model.
$\rho$0 : density of fluid crust.
$\rho$b0 : bulk density of fluid-crust model.
$\xi$0 = $\rho$/$\rho$b0 : ocean-to-bulk density ratio for the fluid-crust model (MW-33)
$\xi$bar0 = $\rho$0/$\rho$b0 : fluid-crust-to-bulk density ratio (MW-33)

# 2. Formulas

## Initialization

Clear function names:

```
Clear[μV, νV, χV, pV, qV, average, μbar, χbar, νbar]
Clear[hRL0, kRL0, fn, pA, pB, An, Bn, hVL0, kVL0, pC, hRLL0, kRLL0]
Clear[xn, lnhn, fμ, fχ, fρ, Λ, Λχ, Λρ, ΛT, tilt, fρL, fχρ, psi]
Clear[ρb0, g0, Kρ, Hρ, kn, hn, knLoad, hnLoad]
Clear[Λωn, Kρr, Hρr, knrDyn, hnrDyn, Dresonant]
Clear[poly, Xa, Xb, Xc, Xd, Xe, k2HC, h2HC, k2DynHC, h2DynHC]
```

Clear function names:

```
Clear[μV, νV, χV, pV, qV, average, μbar, χbar, νbar]
Clear[hRL0, kRL0, fn, pA, pB, An, Bn, hVL0, kVL0, pC, hRLL0, kRLL0]
Clear[xn, lnhn, fμ, fχ, fρ, Λ, Λχ, Λρ, ΛT, tilt, fρL, fχρ, psi]
Clear[ρb0, g0, Kρ, Hρ, kn, hn, knLoad, hnLoad]
Clear[Λωn, Kρr, Hρr, knrDyn, hnrDyn, Dresonant]
Clear[poly, Xa, Xb, Xc, Xd, Xe, k2HC, h2HC, k2DynHC, h2DynHC]
```


## Maxwell rheology

*Remark: 'E' and 'V' refer to 'Elastic' and 'Viscoelastic', respectively.*

Viscoelastic shear modulus $\mu$ in the Fourier domain (MP-123; MW-A.2):

$$\mu V[\mu E\_, \delta\_] := \frac{\mu E}{1 - i\,\delta}$$

Viscoelastic Poisson's ratio $\nu$ in the Fourier domain (MP-125; MW-A.2):

$$\nu V[\nu E\_, \delta\_] := \frac{\nu E - i\,\delta\,(1 + \nu E)/3}{1 - 2\,i\,\delta\,(1 + \nu E)/3}$$

Viscoelastic compressibility factor $\chi$ in the Fourier domain (MW-12):

$$\chi V[\nu E\_, \delta\_] := \frac{1 - 2\,\nu V[\nu E, \delta]}{1 - \nu V[\nu E, \delta]}$$

Viscoelastic parameters (p, q) appearing in viscoelastic-gravitational equations (MW-36):

```
pV[μE_, νE_, δ_] := μV[μE, δ] / (1 - νV[νE, δ])
qV[μE_, νE_, δ_] := pV[μE, νE, δ] * χV[νE, δ]
```

## Two-layer crust with Maxwell rheology:

*Aim: rough model for conductive/convective crust in stagnant lid regime.*
*Elastic parameters (μE,νE) are uniform throughout the whole crust (thickness d).*
*Top layer with relative thickness $f = d_{top} / d$ and Maxwell viscoelasticity parameterized by $\delta_1$.*
*Bottom layer with relative thickness $(1 - f) = d_{bot} / d$ and Maxwell viscoelasticity parameterized by $\delta_2$.*

Weighted average of x (value in top layer) and y (value in bottom layer):

```
average[x_, y_, f_] := f * x + (1 - f) * y
```

Effective shear modulus $\overline{\mu}$ of the membrane (MP-10; MW-42):

```
μbar[μE_, f_, δ1_, δ2_] := average[μV[μE, δ1], μV[μE, δ2], f]
```

Effective Poisson's ratio $\overline{\nu}$ of the membrane (MP-7; MW-43):
*Remark: I set μE=1 in the right-hand side because the ratio does not depend on μE*

```
νbar[νE_, f_, δ1_, δ2_] := average[qV[1, νE, δ1], qV[1, νE, δ2], f] / average[pV[1, νE, δ1], pV[1, νE, δ2], f]
```

Effective compressibility factor $\overline{\chi}$ of the membrane (MW-38):

```
χbar[νE_, f_, δ1_, δ2_] := average[χV[νE, δ1], χV[νE, δ2], f]
```

Effective 'load' compressibility factor $\overline{\chi}_\mu$ of the membrane (MW-48):

```
χμbar[μhat_, νE_, f_, δ1_, δ2_] := average[χV[νE, δ1] / μV[μhat, δ1], χV[νE, δ2] / μV[μhat, δ2], f]
```

## Fluid-crust Love numbers

*Reminder:*
*The fluid-crust model is defined as having the same internal structure as the body to be modeled (called the physical model), except that the crust is fluid and of density ρ0.*
*There are two obvious choices for the fluid-crust density ρ0: it is equal either to the original crust density or to the density of the top layer of the ocean.*
*The latter choice makes it simpler to compute the Love numbers of the fluid-crust model, but it changes the bulk density of the body (except if crust density = ocean density).*

*Method used to obtain the formulas:*
*Incompressible propagator matrix method.*

*Relation between Love numbers:*
*If the surface is fluid, static Love numbers satisfy the hydrostatic relation (MP-35; MW-25):*
*h = k+1*

*Remark: the ' 0' attached to the functions serves to remind us that we are dealing with the fluid-crust model instead of the physical model.*

RL *(for Rigid/Liquid)*: Rigid mantle/Surface ocean (MP-2; MW-C.1):

```
hRL0[ξ0_, n_] := (1 - 3/(2 n + 1) ξ0)^-1
kRL0[ξ0_, n_] := hRL0[ξ0, n] - 1
```

VL *(for Viscoelastic/Liquid)*: Viscoelastic mantle/Surface ocean (MP-118 to 120; MW-C.3 to C.7):

```
(*Auxiliary functions*)
fn[n_] := n / (2 (n - 1) (3 + 4 n + 2 n^2))
pA[y_, ξ0_, n_] := (2 (n - 1) + 3 y^(2 n+1)) (1 - ξ0) + (2 n + 1) y^3 ξ0
pB[y_, ξ0_, n_] := (2 n + 1 - 3 ξ0) (2 (n - 1) (1 - ξ0) + (2 n + 1) y^3 ξ0) - 9 (1 - ξ0) ξ0 y^(2 n+1)
An[y_, ξ0_, n_] := fn[n] * (2 n + 1) * (1 - ξ0) * pA[y, ξ0, n]
Bn[y_, ξ0_, n_] := fn[n] * (1 - ξ0) * pB[y, ξ0, n]

(*Love numbers*)
hVL0[y_, ξ0_, μmHat_, n_] := (An[y, ξ0, n] + (2 n + 1) y^4 μmHat) / (Bn[y, ξ0, n] + (2 n + 1 - 3 ξ0) y^4 μmHat)
kVL0[y_, ξ0_, μmHat_, n_] := hVL0[y, ξ0, μmHat, n] - 1
```

RLL *(for Rigid/Liquid/Liquid)*: Rigid mantle/Ocean/Fluid crust (MW-C.9 to C.11):

```
(*Auxiliary function*)
pC[x_, ξ0_, ξbar0_, n_] := (2 n + 1) (1 - (1 - x^3) ξbar0) + 3 (ξbar0 - ξ0) x^3

(*Love numbers*)
hRLL0[x_, ξ0_, ξbar0_, n_] := (2 n + 1) * (pC[x, ξ0, ξbar0, n] - 3 (ξbar0 - ξ0) * x^(2 n+4)) /
  ((2 n + 1 - 3 ξbar0) * pC[x, ξ0, ξbar0, n] + 9 ξbar0 (ξbar0 - ξ0) * x^(2 n+4))
kRLL0[x_, ξ0_, ξbar0_, n_] := hRLL0[x, ξ0, ξbar0, n] - 1
```

## Relations between Love numbers

Eigenvalue of the operator -(Δ+2) (MW-19):

```
xn[n_] := (n - 1) (n + 2)
```

$l_n - h_n$ relation (MP-25; MW-65):

```
lnhn[ν_, n_] := (1 + ν) / (xn[n] + 1 + ν)
```

Dimensionless coefficients appearing in the $k_n - h_n$ and $k_n' - h_n'$ relations (MW-Table 5):

```
fμ[ν_, n_] := (2 xn[n] (1 + ν)) / (xn[n] + 1 + ν)
fχ[ν_, n_] := - (xn[n] (1 - ν)) / (xn[n] + 1 + ν)
fρ[ν_, n_] := ((n - 1) (n^2 - 3 + (n + 3) ν)) / (xn[n] + 1 + ν)
fρL[ν_, n_] := ((n^2 - 1) (n + 1 - ν)) / (xn[n] + 1 + ν)
fχρ[ν_, n_] := (n (n + 1) (1 - ν)) / (2 (xn[n] + 1 + ν))
```

Membrane spring constant (MP-28; MW-71):

```
Λ[μhat_, ν_, ϵ_, n_] := fμ[ν, n] * μhat * ϵ
```

Compressibility correction (MW-72):

```
Λχ[ν_, χ_, ϵ_, n_] := fχ[ν, n] * χ * ϵ
```

Minor density correction (MW-73):

```
Λρ[δρhat_, ν_, ϵ_, n_] := fρ[ν, n] * δρhat * ϵ
```

Membrane spring constant plus compressibility, density, and dynamical corrections (MW-70):

```
ΛT[μhat_, ν_, χ_, δρhat_, Λω_, ϵ_, n_] :=
 Λ[μhat, ν, ϵ, n] + Λχ[ν, χ, ϵ, n] + Λρ[δρhat, ν, ϵ, n] + Λω
```

Tilt factor (MW-69):

```
tilt[μhat_, ν_, χ_, δρhat_, Λω_, ϵ_, hnValue_, n_] :=
 ΛT[μhat, ν, χ, δρhat, Λω, ϵ, n] * hnValue - (2 n + 1) * δρhat * ϵ
```

Density and compressibility corrections for the $k_n' - h_n'$ relation (MW-79):

```
psi[δρhat_, μhat_, ν_, χ_, χμ_, ϵ_, n_] :=
 (-fχ[ν, n] * χ + fρL[ν, n] * δρhat + χμ/2 + fχρ[ν, n] * (δρhat + χ)^2/μhat) * ϵ
```

## Static Love numbers (gravitational and radial)

Bulk density and surface gravity of the fluid-crust model (MW-85):

```
ρb0[ρb_, δρ0_, ϵ_] := ρb - 3 * δρ0 * ϵ
g0[g_, ρb_, δρ0_, ϵ_] := g * (1 - 3 * δρ0/ρb * ϵ)
```

Density corrections in the Love number formulas (MW-99):

```
Kρ[δρhat0_, ϵ_, hn0_, n_] := 2 * ((n - 1) hn0 - (2 n + 1)) * δρhat0 * ϵ
Hρ[δρhat_, δρhat0_, ν_, ϵ_, ξ0_, hn0_, n_] :=
 Λρ[δρhat, ν, ϵ, n] - (2 n + 1)/hn0 * δρhat * ϵ + 3 ξ0/(2 n + 1) * Kρ[δρhat0, ϵ, hn0, n]
```

Formulas for gravitational and radial Love numbers (MW-98):
*Assumption: the fluid-crust density is equal to the density of the top layer of the ocean: ρ0=ρ*

```
kn[μhat_, ν_, χ_, δρhat_, ϵ_, ξ0_, hn0_, n_] :=
 hn0 * (1 + 3 ξ0/(2 n + 1) * (hn0 * Λ[μhat, ν, ϵ, n]/(1 + Λ[μhat, ν, ϵ, n]) + Kρ[δρhat, ϵ, hn0, n]))^-1 - 1
hn[μhat_, ν_, χ_, δρhat_, ϵ_, ξ0_, hn0_, n_] :=
 hn0 * (1 + (1 + 3 ξ0/(2 n + 1) * hn0) * Λ[μhat, ν, ϵ, n] +
     Λχ[ν, χ, ϵ, n] + Hρ[δρhat, δρhat, ν, ϵ, ξ0, hn0, n])^-1
```

Load Love numbers (MW-80 and 81):

```
knLoad[μhat_, ν_, χ_, δρhat_, ϵ_, ξ0_, hn0_, n_] := -1 +
  ΛT[μhat, ν, χ, δρhat, 0, ϵ, n] * hn[μhat, ν, χ, δρhat, ϵ, ξ0, hn0, n] - (2 n + 1) * δρhat * ϵ
hnLoad[μhat_, ν_, χ_, χμ_, δρhat_, ϵ_, ξ_, ξ0_, hn0_, n_] :=
 (ΛT[μhat, ν, χ, δρhat, 0, ϵ, n] * hn[μhat, ν, χ, δρhat, ϵ, ξ0, hn0, n] -
    (2 n + 1)/(3 ξ) * (1 + psi[δρhat, μhat, ν, χ, χμ, ϵ, n])) / (1 + ΛT[μhat, ν, χ, δρhat, 0, ϵ, n])
```

## Dynamical Love numbers (gravitational and radial)

Dynamical correction if rigid mantle model (MW-116):

```
Λωn[qω_, z_, n_] := - qω/(n (n + 1)) * (n + 1 + n * z^(2 n + 1))/(1 - z^(2 n + 1))
```

Density corrections in the dynamical Love number formulas, assuming the rigid mantle model (MW-120):

```
Kρr[δρhat_, ϵ_, ξ0_, n_] := 2 * ((n - 1) * 1/(1 - 3 ξ0 / (2 n + 1)) - (2 n + 1)) * δρhat * ϵ
Hρr[δρhat_, ν_, ϵ_, ξ0_, n_] := Kρr[δρhat, ϵ, ξ0, n] + Λρ[δρhat, ν, ϵ, n] + 3 (1 - ξ0) δρhat * ϵ
```

Formulas for the dynamical gravitational and radial Love numbers, assuming the rigid mantle model (MW-119):

$$\text{knrDyn}[\mu\text{hat\_}, \nu\_, \chi\_, \delta\rho\text{hat\_}, \epsilon\_, q\omega\_, z\_, \xi 0\_, n\_] := \frac{1}{1-3\,\xi 0/(2\,n+1)} * \left(1+\frac{3\,\xi 0}{2\,n+1} * \frac{1}{1+\Lambda[\mu\text{hat}, \nu, \epsilon, n]+\Lambda\omega n[q\omega, z, n]} * \left(\frac{\Lambda[\mu\text{hat}, \nu, \epsilon, n]+\Lambda\omega n[q\omega, z, n]}{1-3\,\xi 0/(2\,n+1)}+K\rho r[\delta\rho\text{hat}, \epsilon, \xi 0, n]\right)\right)^{-1} - 1$$

$$\text{hnrDyn}[\mu\text{hat\_}, \nu\_, \chi\_, \delta\rho\text{hat\_}, \epsilon\_, q\omega\_, z\_, \xi 0\_, n\_] := \frac{1}{1-3\,\xi 0/(2\,n+1)} * \left(1+\frac{\Lambda[\mu\text{hat}, \nu, \epsilon, n]+\Lambda\omega n[q\omega, z, n]}{1-3\,\xi 0/(2\,n+1)}+\Lambda\chi[\nu, \chi, \epsilon, n]+H\rho r[\delta\rho\text{hat}, \nu, \epsilon, \xi 0, n]\right)^{-1}$$

Approximate resonant ocean thickness (MW-123):

*Remark: neglect crust-ocean density contrast and compressibility factor, but keep the dependence of $\Lambda$ on $\nu$ because it can be significant if $\Lambda$ is large*

$$\text{Dresonant}[R\_, q\omega\_, \xi\_, \mu\text{hat\_}, \nu\_, \epsilon\_, n\_] := R * \frac{q\omega}{n\,(n+1)} * \left(1-\frac{3\,\xi}{2\,n+1}+\text{Re}[\Lambda[\mu\text{hat}, \nu, \epsilon, n]]\right)^{-1}$$

## ■ A simple thick shell model: the homogeneous crust model

*Reference: MP-Appendix A and MW-Appendix E*

*Characteristics: homogeneous and incompressible crust, no density contrast at the crust-ocean boundary.*

*In the static case, one can solve the model in terms of fluid-crust Love numbers. which means that the structure below the crust can be freely chosen (MP-Appendix A).*

*In the dynamical case, an explicit solution can be obtained for the rigid mantle model: infinitely rigid mantle + homogeneous and incompressible ocean (MW-Appendix E).*

Geometrical factors appearing in the Love number relations (MW-Table 11):

*Recall that $x = \frac{R-d}{R}$ is the relative radius of the crust-ocean boundary*

$$\text{poly}[x\_] := 24+40\,x^3-45\,x^7-19\,x^{10}$$

$$\text{Xa}[x\_] := \frac{24}{5}\left(19-75\,x^3+112\,x^5-75\,x^7+19\,x^{10}\right)/\text{poly}[x]$$

$$\text{Xb}[x\_] := x^2\left(19+45\,x^3-40\,x^7-24\,x^{10}\right)/\text{poly}[x]$$

$$\text{Xc}[x\_] := \frac{1}{5}\left(36-100\,x^3+308\,x^5-225\,x^7-19\,x^{10}\right)/\text{poly}[x]$$

$$\text{Xd}[x\_] := \frac{1}{2} * x^2\left(3+5\,x^3-10\,x^7+2\,x^{10}\right)/\text{poly}[x]$$

$$\text{Xe}[x\_] := 5\,x^2\left(1-x^3-x^7+x^{10}\right)/\text{poly}[x]$$

Factors appearing in the $k_2 - h_2$ and $l_2 - h_2$ formulas (MW-E.3):

$$\text{zh}[\mu\text{hat\_}, \Lambda\omega\_, x\_] := \frac{\text{Xa}[x] * \mu\text{hat}+\text{Xb}[x] * \Lambda\omega}{\mu\text{hat}+\text{Xe}[x] * \Lambda\omega}$$

$$\text{zl}[\mu\text{hat\_}, \Lambda\omega\_, x\_] := \frac{\text{Xc}[x] * \mu\text{hat}+\text{Xd}[x] * \Lambda\omega}{\mu\text{hat}+\text{Xe}[x] * \Lambda\omega}$$

Static Love numbers of degree 2 in terms of fluid-crust Love number h20 (MP-112):

$$\text{k2HC}[\xi\_, \mu\text{hat\_}, x\_, \text{h20\_}] := \text{h20} * \left(1+\frac{3}{5} * \xi * \text{h20} * \frac{\text{Xa}[x] * \mu\text{hat}}{1+\text{Xa}[x] * \mu\text{hat}}\right)^{-1} - 1$$

$$\text{h2HC}[\xi\_, \mu\text{hat\_}, x\_, \text{h20\_}] := \text{h20} * \left(1+\left(1+\frac{3}{5} * \xi * \text{h20}\right) * \text{Xa}[x] * \mu\text{hat}\right)^{-1}$$

Dynamical Love numbers of degree 2 if rigid mantle model (MW-E.7):

```
k2DynHC[ξ_, μhat_, Λω_, x_] := 3/5 ξ * 1/(1 - 3/5 * ξ + zh[μhat, Λω, x] * μhat)

h2DynHC[ξ_, μhat_, Λω_, x_] := 1/(1 - 3/5 * ξ + zh[μhat, Λω, x] * μhat)
```

# Part III: Other examples

Most examples are taken from papers MP and MW, to which the reader is referred for more explanations.

## 0. Constants

Set plotting options:

```
SetOptions[{Plot, LogLinearPlot}, PlotStyle → {Thick, Black}, Frame → True,
  FrameStyle → Directive[Thickness[0.003]], ImageSize → 280, AspectRatio → 1,
  Axes → None, BaseStyle → {14, FontFamily → "Arial"}, PlotRange → All];
```

Clear variables that are left unspecified in this notebook:

```
Clear[μ, ν, χ, μE, νE, χE, fϵ, δ1, δ2, μEm, μEhat]
Clear[ρb, ρ, ρbar, δρ, δρhat]
Clear[ω, Rm, R, g, n]
```

Units:

*radius: m*

*density:* $kg / m^3$

*surface gravity:* $m / s^2$

*shear modulus: Pa*

*angular frequency:* $s^{-1}$

Bulk and orbital parameters of Europa (MW-Table 2):

```
ωEuropa = 2.048 * 10^-5;
RadiusEuropa = 1560.8 * 10^3;
BulkDensityEuropa = 3013.;
SurfaceGravityEuropa = 1.315;
MantleRadiusEuropa = RadiusEuropa - 170. * 10^3(*MW-caption of Table 6*);
ruleEuropa = {ω → ωEuropa, Rm → MantleRadiusEuropa, R → RadiusEuropa,
   ρb → BulkDensityEuropa, g → SurfaceGravityEuropa, μEm -> μEmantle, n → 2};
qωEuropa = (ω^2 * R)/g /. ruleEuropa

0.00049783
```

Bulk and orbital parameters of Titan (MW-Table 2):

```
ωTitan = 0.456 * 10^-5;
RadiusTitan = 2574.76 * 10^3;
BulkDensityTitan = 1881.5;
SurfaceGravityTitan = 1.354;
MantleRadiusTitan = RadiusTitan - 350. * 10^3(*MW-caption of Table 6*);
ruleTitan = {ω → ωTitan, Rm → MantleRadiusTitan, R → RadiusTitan,
   ρb → BulkDensityTitan, g → SurfaceGravityTitan, μEm -> μEmantle, n → 2};
qωTitan = (ω^2 * R)/g /. ruleTitan

0.000039541
```

Elastic parameters (MP-Table 3 and MW-Table 7):

```
μEmantle = 40. * 10^9;
μEmantleSoft = 2. * 10^8;
μEice = 3.5 * 10^9;
νEice = 0.33;
χEice = (1 - 2 νEice)/(1 - νEice);
ruleElastic = {μ → μEice, ν -> νEice, χ -> χEice}
```

$\{\mu \to 3.5\times 10^{9},\ \nu \to 0.33,\ \chi \to 0.507463\}$

Density of pure water:

```
ρWater = 1000.;
```

MP: Replacement rules and rheology models (MP-Table 4)

```
μEicehat = μEice/(ρWater * RadiusEuropa * SurfaceGravityEuropa);
ruleMP1 = {μEhat -> μEicehat, νE → νEice, fϵ → 0.4, n -> 2};
ruleMP2 = {{δ1 -> 10^-7, δ2 → 0.01}, {δ1 -> 10^-7, δ2 → 1}, {δ1 -> 10^-7, δ2 → 100}};
μMP = μbar[μEhat, fϵ, δ1, δ2] /. ruleMP1 /. ruleMP2
νMP = νbar[νE, fϵ, δ1, δ2] /. ruleMP1 /. ruleMP2
χMP = χbar[νE, fϵ, δ1, δ2] /. ruleMP1 /. ruleMP2
```

{1.70518 + 0.0102307 i, 1.19369 + 0.511583 i, 0.682213 + 0.0102307 i}

{0.330011 - 0.0009043 i, 0.384862 - 0.0337109 i, 0.330154 + 0.0034113 i}

{0.507449 + 0.00201463 i, 0.414746 + 0.140121 i, 0.203055 + 0.00460047 i}

MW: Density models for the crust of Europa and Titan (MW-Table 6):

```
{ρI, ρO} = {930., 1020.};
ruleL = {ρbar → ρI, ρ → ρO, δρ → ρI - ρO, δρhat → (ρI - ρO)/ρO}
{ρI, ρO} = {930., 1280.};
ruleM = {ρbar → ρI, ρ → ρO, δρ → ρI - ρO, δρhat → (ρI - ρO)/ρO}
{ρI, ρO} = {1167., 1280.};
ruleD = {ρbar → ρI, ρ → ρO, δρ → ρI - ρO, δρhat → (ρI - ρO)/ρO}
ruleLMD = {ruleL, ruleM, ruleD};
```

{ρbar → 930., ρ → 1020., δρ → -90., δρhat → -0.0882353}

{ρbar → 930., ρ → 1280., δρ → -350., δρhat → -0.273438}

{ρbar → 1167., ρ → 1280., δρ → -113., δρhat → -0.0882813}

MW: Rheology models for Europa’s crust (MW-Tables 7 and 8):

```
{δ1E, δ1C, δ1F} = {1, 1, 1} * 10^-7;
{δ2E, δ2C, δ2F} = {0.1, 1., 10.};
{fE, fC, fF} = {0.4, 0.4, 0.1};
ruleE = {μE → μEice, νE -> νEice, fϵ -> fE, δ1 -> δ1E, δ2 -> δ2E};
ruleC = {μE → μEice, νE -> νEice, fϵ -> fC, δ1 -> δ1C, δ2 -> δ2C};
ruleF = {μE → μEice, νE -> νEice, fϵ -> fF, δ1 -> δ1F, δ2 -> δ2F};
ruleECF = {ruleE, ruleC, ruleF};
```

The corresponding effective viscoelastic parameters are those of MW-Table 8:

```
μECF = μbar[μE, fϵ, δ1, δ2] /. ruleECF;
νECF = νbar[νE, fϵ, δ1, δ2] /. ruleECF;
χECF = χbar[νE, fϵ, δ1, δ2] /. ruleECF;
TableForm[Transpose@{μECF / μEice, νECF, χECF},
 TableHeadings → {{"Elastic-like", "Critical", "Fluid-like"}, {"μ̄/μE", "ν̄", "χ̄"}}]
```

| | $\overline{\mu}/\mu$E | $\overline{\nu}$ | $\overline{\chi}$ |
|---|---|---|---|
| Elastic-like | 0.994059 + 0.059406 i | 0.331073 − 0.00893931 i | 0.506135 + 0.0200592 i |
| Critical | 0.7 + 0.3 i | 0.384862 − 0.0337109 i | 0.414746 + 0.140121 i |
| Fluid-like | 0.108911 + 0.0891089 i | 0.435049 + 0.0611982 i | 0.0609446 + 0.0674813 i |

The associated replacement rules are:

```
ruleEbar = {μ -> μECF[[1]], ν -> νECF[[1]], χ -> χECF[[1]]};
ruleCbar = {μ -> μECF[[2]], ν -> νECF[[2]], χ -> χECF[[2]]};
ruleFbar = {μ -> μECF[[3]], ν -> νECF[[3]], χ -> χECF[[3]]};
```

# 1. Effective viscoelastic parameters

## Example 1.1: viscoelastic parameters of a uniform layer with Maxwell rheology (MP-Fig 1)

*Recall that $\delta = \mu_E / (\omega * \eta)$ is a dimensionless parameter inversely proportional to tidal frequency $\omega$ and viscosity $\eta$;*
*$\delta$ quantifies viscous effects in Maxwell rheology.*

Here are the formulas for the viscoelastic shear modulus $\mu$, Poisson's ratio $\nu$, and compressibility factor $\chi$ in the Fourier domain ('E' refers to 'Elastic' and 'V' to 'Viscoelastic'):

```
{μV[μE, δ], νV[νE, δ], χV[νE, δ]} // Simplify
```

$$\left\{\frac{i\,\mu E}{i+\delta}, \frac{\delta + 3\,i\,\nu E + \delta\,\nu E}{3\,i + 2\,\delta\,(1+\nu E)}, \frac{3\,i - 6\,i\,\nu E}{3\,i + \delta - 3\,i\,\nu E + \delta\,\nu E}\right\}$$

Plot the viscoelastic parameters ($\mu$, $\chi$, $\nu$) as functions of $\delta$ (MP-Fig 1).
*Remark: the shear modulus is divided its elastic value.*




## ■ Example 1.2: effective viscoelastic parameters of a two-layer crust (MP-Fig 2)

*In this example, elastic parameters ($\mu$,$\nu$) are uniform throughout the whole crust (thickness d); Top layer with relative thickness $f = d_{top} / d$ and Maxwell viscoelasticity parameterized by $\delta_1$. Bottom layer with relative thickness $(1 - f) = d_{bot} / d$ and Maxwell viscoelasticity parameterized by $\delta_2$.*

The top layer thickness is 0.4 times the total thickness:

```
fd = 0.4;
```

The viscosity of the bottom layer is $10^6$ smaller than the viscosity of the top layer, so that $\delta_2 = \text{mult} * \delta_1$ with:

```
mult = 10^6;
```

Now, plot effective viscoelastic parameters ($\overline{\mu}$, $\overline{\chi}$, $\overline{\nu}$) as functions of $\delta_1$ (MP-Fig 2):




## ■ Example 1.3: effective viscoelastic parameters of a two-layer crust (MW-Fig 3)

*As in Example 1.2, except that the top layer remains purely elastic while the $\delta_2$ parameter varies.*

The top layer thickness is 0.4 times the total thickness:

```
fd = 0.4;
```

The top layer is elastic so $\delta_1$=0:

```
δ1elastic = 0;
```

Critical transitions for $\chi$ and $\nu$ (MW-Section 3.4):

$$\{\delta p,\ \delta pp\} = \left\{3 * \frac{1-\nu}{1+\nu},\ \sqrt{\frac{3}{fd} * \frac{1-\nu}{1+\nu}}\right\} /.\ ruleElastic$$

```
{1.51128, 1.94376}
```

Now, plot effective viscoelastic parameters ($\overline{\mu}$, $\overline{\chi}$, $\overline{\nu}$) as functions of $\delta_2$ (MW-Fig 3):










For completeness, plot the effective 'load' compressibility parameter $\overline{\chi}_\mu$ as a function of $\delta_2$ (this figure is not in the paper MW);
assume that $\mu_E / (\rho g R) = 1$:



# 2. $l_2 - h_2$ relation

## Example 2.1: $l_2 - h_2$ ratio as a function of the inverse frequency/inverse viscosity of the crust (MP-Fig 3)

The crust is divided into two layers, the top layer thickness being 0.4 times the total thickness:

```
fd = 0.4;
```

The viscosity of the bottom layer is $10^6$ smaller than the viscosity of the top layer, so that $\delta_2 = \text{mult} * \delta_1$ with:

```
mult = 10*^6;
```

Now, plot the $l_2 / h_2$ ratio as a function of $\delta_1$ (MP-Fig 3):




## Example 2.2: $l_2 - h_2$ ratio as a function of crust thickness (MP-Fig 4)

The membrane formula for the $l_2 / h_2$ ratio is:

```
lnhn[ν, 2]
```

$$\frac{1+\nu}{5+\nu}$$

First, compute the membrane approximation of the $l_2 / h_2$ ratio for the rheology models (H,E,C,F) of the paper MP.

In MP and MW, the membrane approximation of $l_2 / h_2$ is only computed at order 1 in ϵ=d/R.

```
zlH = lnhn[1 / 2, 2];
zlE = lnhn[νMP[[1]], 2];
zlC = lnhn[νMP[[2]], 2];
zlF = lnhn[νMP[[3]], 2];
{zlH, zlE, zlC, zlF}
```

$\left\{\frac{3}{11}, 0.249533 - 0.000127326\, i, 0.257206 - 0.00465012\, i, 0.249553 + 0.000480286\, i\right\}$

For comparison, consider $l_2 / h_2$ ratio of the homogeneous crust model.
This model assumes that the crust is homogeneous and incompressible, and is not appropriate for the 2-layer rheology.
The static $l_2 / h_2$ ratio for this model is given by zl[1,0,x]=Xc[x]

```
zl[1, 0, x]
```

$$\frac{36 - 100\, x^3 + 308\, x^5 - 225\, x^7 - 19\, x^{10}}{5\, (24 + 40\, x^3 - 45\, x^7 - 19\, x^{10})}$$

Finally, plot the $l_2 / h_2$ ratio as a function of crust thickness for a two-layer crust (MP-Fig 4).




# 3. $k_2 - h_2$ relation

## ■ Example 3.1: normalized tilt factor in function of the inverse frequency/inverse viscosity of the crust (MP-Fig 5)

For safety, clear variables to be replaced:

```
Clear[ν, χ, δρhat, Λω, ϵ, n, μEhat, νE, fϵ, δ1, δ2];
```

Assume static tides (Λ$\omega$=0) of degree 2 and crust density=ocean density=1000 kg/m3 (MP-Table 3);
Moreover there is no compressibility factor in the paper MP: $\chi$=0.
Under these assumptions, the tilt factor divided by $h_2$ (called normalized tilt factor, see MP-37) is given by

```
ΛT[μ̂, ν, χ, δρhat, Λω, ϵ, n] /. {χ → 0, δρhat → 0, Λω → 0, n → 2}
```

$$\frac{8\, \epsilon\, (1 + \nu)\, \hat{\mu}}{5 + \nu}$$

which is equal to Λ:

```
Λ[μ̂, ν, ϵ, n] /. {n → 2}
```

$$\frac{8\,\epsilon\,(1+\nu)\,\hat{\mu}}{5+\nu}$$

First, define replacement rules for rheology model of Europa's crust (MP-Table 3)

```
μEicehat = μEice / (ρWater * RadiusEuropa * SurfaceGravityEuropa);
mult = 10^6;(*the viscosities of the two crust layers are in a constant ratio*)
ruleMPfig5 = {μEhat -> μEicehat, νE → νEice, fϵ -> fE, δ2 -> mult * δ1,
   δρhat → 0, Δω → 0, ϵ → 1, n → 2}(*ϵ=1 because Λ is in units of d/R*);
```

Next, define Λ as a one-variable function just for this plot:

```
Clear[ΛMP];
ΛMP[δ1_] = Λ[μbar[μEhat, fϵ, δ1, δ2], νbar[νE, fϵ, δ1, δ2], ϵ, n] /. ruleMPfig5;
```

For comparison, plot also the curves for $\Lambda_T$ , i.e. with the correct compressibility correction ($\chi \neq 0$).

For this purpose, define $\Lambda_T$ as a one-variable function just for this plot:

```
Clear[ΛMPcomp];
ΛMPcomp[δ1_] = ΛT[μbar[μEhat, fϵ, δ1, δ2],
     νbar[νE, fϵ, δ1, δ2], χbar[νE, fϵ, δ1, δ2], δρhat, Δω, ϵ, n] /. ruleMPfig5;
```

Finally, plot the normalized tilt factor as a function of $\delta_1$ (MP-Figure 5):







## ■ Example 3.2: normalized tilt factor as a function of crust thickness (MP-Fig 6)

Same assumptions as for Example 3.1.
Plot normalized tilt factor as a function of crust thickness for different crustal rheology models.
For comparison, plot also the curves for $\Lambda_T$ , i.e. with the correct compressibility correction ($\chi \neq 0$):




## Example 3.3: slope of tilt factor at zero crust thickness (MW-Fig 7)

For safety, clear variables to be replaced:

```
Clear[ν, χ, Δω, ϵ, n];
Clear[Rm, R, ρ, ρb, g, μEm];
```

Here are the formulas for the degree-two dimensionless coefficients appearing in the tilt factor formula (MW-Table 5):

```
{fμ[ν, n], fχ[ν, n], fρ[ν, n]} /. {n → 2}
```

$$\left\{\frac{8\,(1+\nu)}{5+\nu},\ -\frac{4\,(1-\nu)}{5+\nu},\ \frac{1+5\,\nu}{5+\nu}\right\}$$

Here is the formula for the degree-two membrane spring constant modified by corrections (MW-70) :

```
ΛT[μ̂, ν, χ, δρ̂, Δω, ϵ, n] /. {n → 2}
```

$$\Lambda\omega-\frac{4\,\epsilon\,(1-\nu)\,\chi}{5+\nu}+\frac{\epsilon\,(1+5\,\nu)\,\hat{\delta\rho}}{5+\nu}+\frac{8\,\epsilon\,(1+\nu)\,\hat{\mu}}{5+\nu}$$

First, compute the fluid-crust Love numbers at zero crust thickness with the VL model (MW-Figure 7, see caption):
Remark: since the crust is of zero thickness, the fluid-crust model and the physical model have the same bulk density.

```
h20 = hVL0[Rm/R, ρ/ρb, μEm/(ρb * g * R), n] /. ruleEuropa;
{h20L, h20M, h20D} = h20 /. ruleLMD
```

```
{1.2696, 1.35484, 1.35484}
```

Next, compute the various approximations for the slope (at zero crust thickness) of Europa’s degree-two tilt factor, for the elastic-like rheology model:

```
approx = {0, 0, 0, 0};
approx[[1]] =
  First@Limit[(ΔT[μECF[[1]]/(ρ * g * R), νECF[[1]], 0, 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[2]] = First@Limit[
    (ΔT[μECF[[1]]/(ρ * g * R), νECF[[1]], χECF[[1]], 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[3]] = First@Limit[(ΔT[μECF[[1]]/(ρ * g * R), νECF[[1]], χECF[[1]], δρhat, 0, ϵ, n] * h20/ϵ) /.
      ruleEuropa, {ϵ → 0}];
approx[[4]] = First@Limit[(tilt[μECF[[1]]/(ρ * g * R), νECF[[1]], χECF[[1]], δρhat, 0, ϵ, h20, n] *
        1/ϵ) /. ruleEuropa, {ϵ → 0}];
caseE = approx /. ruleLMD;
```

Do the same for the critical rheology model:

```
approx = {0, 0, 0, 0};
approx[[1]] =
  First@Limit[(ΔT[μECF[[2]]/(ρ * g * R), νECF[[2]], 0, 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[2]] = First@Limit[
    (ΔT[μECF[[2]]/(ρ * g * R), νECF[[2]], χECF[[2]], 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[3]] = First@Limit[(ΔT[μECF[[2]]/(ρ * g * R), νECF[[2]], χECF[[2]], δρhat, 0, ϵ, n] * h20/ϵ) /.
      ruleEuropa, {ϵ → 0}];
approx[[4]] = First@Limit[(tilt[μECF[[2]]/(ρ * g * R), νECF[[2]], χECF[[2]], δρhat, 0, ϵ, h20, n] *
        1/ϵ) /. ruleEuropa, {ϵ → 0}];
caseC = approx /. ruleLMD;
```

Do the same for the fluid-like rheology model:

```
approx = {0, 0, 0, 0};
approx[[1]] =
  First@Limit[(ΔT[μECF[[3]]/(ρ*g*R), νECF[[3]], 0, 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[2]] = First@Limit[
    (ΔT[μECF[[3]]/(ρ*g*R), νECF[[3]], χECF[[3]], 0, 0, ϵ, n] * h20/ϵ) /. ruleEuropa, {ϵ → 0}];
approx[[3]] = First@Limit[(ΔT[μECF[[3]]/(ρ*g*R), νECF[[3]], χECF[[3]], δρhat, 0, ϵ, n] * h20/ϵ) /.
      ruleEuropa, {ϵ → 0}];
approx[[4]] = First@Limit[(tilt[μECF[[3]]/(ρ*g*R), νECF[[3]], χECF[[3]], δρhat, 0, ϵ, h20, n] *
         1/ϵ) /. ruleEuropa, {ϵ → 0}];
caseF = approx /. ruleLMD;
```

As the instructions for Figure 7 of MW are rather long, I give instead the numerical values:

```
Print["ELASTIC-LIKE: LEFT PANELS OF MW-FIG 7"]
Style[TableForm[caseE, TableHeadings → {{"Light", "Mixed", "Dense"}, None}], 11]
Print["CRITICAL: MIDDLE PANELS OF MW-FIG 7"]
Style[TableForm[caseC, TableHeadings → {{"Light", "Mixed", "Dense"}, None}], 11]
Print["FLUID-LIKE: RIGHT PANELS OF MW-FIG 7"]
Style[TableForm[caseF, TableHeadings → {{"Light", "Mixed", "Dense"}, None}], 11]
```

ELASTIC-LIKE: LEFT PANELS OF MW-FIG 7

| | | | | |
|---|---|---|---|---|
| Light | 4.21586 + 0.230631 i | 3.89354 + 0.212998 i | 3.83774 + 0.213844 i | 4.27891 + 0.213844 i |
| Mixed | 3.58505 + 0.196122 i | 3.24109 + 0.177306 i | 3.05656 + 0.180102 i | 4.42375 + 0.180102 i |
| Dense | 3.58505 + 0.196122 i | 3.24109 + 0.177306 i | 3.18152 + 0.178209 i | 3.62292 + 0.178209 i |

CRITICAL: MIDDLE PANELS OF MW-FIG 7

| | | | | |
|---|---|---|---|---|
| Light | 3.08094 + 1.25498 i | 2.84539 + 1.15903 i | 2.78453 + 1.16215 i | 3.22571 + 1.16215 i |
| Mixed | 2.61995 + 1.0672 i | 2.36858 + 0.964809 i | 2.16733 + 0.975145 i | 3.53452 + 0.975145 i |
| Dense | 2.61995 + 1.0672 i | 2.36858 + 0.964809 i | 2.30361 + 0.968146 i | 2.74502 + 0.968146 i |

FLUID-LIKE: RIGHT PANELS OF MW-FIG 7

| | | | | |
|---|---|---|---|---|
| Light | 0.475935 + 0.414983 i | 0.439548 + 0.383256 i | 0.374039 + 0.377686 i | 0.815215 + 0.37768 |
| Mixed | 0.404722 + 0.35289 i | 0.365892 + 0.319033 i | 0.149254 + 0.300616 i | 1.51644 + 0.300616 |
| Dense | 0.404722 + 0.35289 i | 0.365892 + 0.319033 i | 0.295949 + 0.313087 i | 0.737355 + 0.31308 |

# 4. Static Love numbers

## ■ Example 4.1: Radial Love number of Europa as a function of the inverse frequency/inverse viscosity of the crust, for diurnal tides (MP-Fig 7A)

There is no density contrast between crust and ocean, so that the fluid-crust Love numbers are computed with the bulk density and surface gravity of the physical model.

For safety, clear variables to be replaced:

```
Clear[ξ, μEhat, νE, fϵ, δ1, δ2, δρhat, Δω, ϵ, n, hn0];
Clear[Rm, R, ρ, ρb, g, hn0];
```

First, define the replacement rules for the Europa model considered in the MP paper (MP-Tables 2, 3, and 4):

```
mult = 10^6;(*the viscosities of the two crust layers are in a constant ratio*)
d0 = 20. * 10^3;(*the crust is 20 km thick*)
μEicehat = μEice / (ρWater * RadiusEuropa * SurfaceGravityEuropa);
(*nondimensional shear modulus of elastic ice*)
ruleMPfig7 = {ξ -> ρWater / BulkDensityEuropa, μEhat -> μEicehat, νE → νEice,
   fϵ -> fE, δ2 -> mult * δ1, δρhat → 0, Λω → 0, ϵ → d0 / RadiusEuropa, n → 2};
```

Next, compute the fluid-crust Love numbers for diurnal tides: elastic mantle (h20E) or rigid mantle (h20R):

```
h20E = hVL0[Rm / R, ρ / ρb, μEm / (ρb * g * R), n] /. ruleEuropa /. {ρ → ρWater}
h20R = hRL0[ρ / ρb, n] /. ruleEuropa /. {ρ → ρWater}
```

```
1.26351
```

```
1.24865
```

Then, express the radial Love number as a function of $\delta_1$ and another variable called $\chi$fac:
I leave the possibility to include ($\chi$fac=1) or exclude ($\chi$fac=0) the compressibility factor:

```
Clear[h2E, h2R];
h2E[δ1_, χfac_] =
  hn[μbar[μEhat, fϵ, δ1, δ2], νbar[νE, fϵ, δ1, δ2], χfac * χbar[νE, fϵ, δ1, δ2],
     δρhat, ϵ, ξ, hn0, n] /. ruleMPfig7 /. {hn0 → h20E};
h2R[δ1_, χfac_] = hn[μbar[μEhat, fϵ, δ1, δ2], νbar[νE, fϵ, δ1, δ2],
     χfac * χbar[νE, fϵ, δ1, δ2], δρhat, ϵ, ξ, hn0, n] /. ruleMPfig7 /. {hn0 → h20R};
```

Finally, plot radial Love number for diurnal tides:







## ▪ Example 4.2: Radial Love number of Europa as a function of the inverse frequency/inverse viscosity of the crust, for NSR tides (MP-Fig 7B)

The replacement rule is the same as in Example 4.1.

First, compute the fluid-crust Love numbers for NSR tides (due to nonsynchronous rotation); In comparison with diurnal tides, the shear modulus of the mantle is reduced: consider either a soft mantle with $\mu$=0.2 GPa (h20S) or a completely fluid mantle (h20F):

```
Clear[μEmNSR];
h20S =
  hVL0[Rm/R, ρ/ρb, μEmNSR/(ρb * g * R), n] /. ruleEuropa /. {ρ → ρWater} /. {μEmNSR -> μEmantleSoft};
h20F = hVL0[Rm/R, ρ/ρb, 0, n] /. ruleEuropa /. {ρ → ρWater};
```

Next, express the radial Love number as a function of $\delta_1$ and another variable called $\chi$fac: I leave the possibility to include ($\chi$fac=1) or exclude ($\chi$fac=0) the compressibility factor:

```
Clear[h2S, h2F];
h2S[δ1_, χfac_] =
  hn[μbar[μEhat, f∊, δ1, δ2], νbar[νE, f∊, δ1, δ2], χfac * χbar[νE, f∊, δ1, δ2],
     δρhat, ∊, ξ, hn0, n] /. ruleMPfig7 /. {hn0 → h20S};
h2F[δ1_, χfac_] = hn[μbar[μEhat, f∊, δ1, δ2], νbar[νE, f∊, δ1, δ2],
     χfac * χbar[νE, f∊, δ1, δ2], δρhat, ∊, ξ, hn0, n] /. ruleMPfig7 /. {hn0 → h20F};
```

Finally, plot radial Love number for NSR tides:







## ■ Example 4.3: Tidal Love numbers of Titan as functions of crust thickness for different crust/ocean density models (MW-Fig 5)

For safety, clear variables to be replaced:

```
Clear[Rm, R, g, ∊, hn0, n]
Clear[ρ, ρb, δρ, δρhat]
Clear[μ, ν, χ, μEm]
```

First, evaluate partially the fluid-crust Love number h20 with the VL model, using the replacement rule ‘ruleTitan’;
the core-mantle layer is assumed to be elastic;
don’t forget to use the fluid-crust bulk density and gravity;
h20 now depends on the three variables ($\rho$, $\delta\rho$, $\epsilon$):

```
h20 = hVL0[Rm/R, ρ/ρb0[ρb, δρ, ϵ], μEm/(ρb0[ρb, δρ, ϵ] * g0[g, ρb, δρ, ϵ] * R), n] /. ruleTitan;
```

Next, define the gravitational and radial Love numbers (degree 2) of Titan as functions of crust thickness;
As the crust is elastic, use the replacement rule ‘ruleElastic’.
Define $k_n$and $h_n$ functions for the three density models (Light, Mixed, Dense), using the replacement rule ‘ruleLMD’:

```
Clear[knL, knM, knD, hnL, hnM, hnD]
{knL[ϵ_], knM[ϵ_], knD[ϵ_]} =
  kn[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. ruleTitan /. ruleElastic /.
    {hn0 → h20} /. ruleLMD;
{hnL[ϵ_], hnM[ϵ_], hnD[ϵ_]} = hn[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /.
      ruleTitan /. ruleElastic /. {hn0 → h20} /. ruleLMD;
```

Plot the Love numbers as functions of crust thickness (MW-Fig 5):




## ▪ Example 4.4: Tidal Love numbers of Titan in function of crust thickness: choose another fluid-crust density (not in MW)

Similar to Example 4.3, except that the mantle is infinitely rigid.
Compare the results with two different fluid-crust densities: either $\rho0=\rho$ or $\rho0=\rho$bar

For safety, clear variables to be replaced:

```
Clear[R, Rm, g, ϵ, hn0, n]
Clear[ρ, ρb, δρ, δρhat]
Clear[μ, ν, χ]
```

As in Example 4.3, define $k_n$and $h_n$ functions for the three density models, assuming that $\rho0=\rho$ (the mantle is infinitely rigid):

```
h20R = hRL0[ρ/ρb0[ρb, δρ, ϵ], n] /. ruleTitan;
Clear[knLR, knMR, knDR, hnLR, hnMR, hnDR]
{knLR[ϵ_], knMR[ϵ_], knDR[ϵ_]} =
  kn[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. ruleTitan /. ruleElastic /.
    {hn0 → h20R} /. ruleLMD;
{hnLR[ϵ_], hnMR[ϵ_], hnDR[ϵ_]} = hn[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /.
      ruleTitan /. ruleElastic /. {hn0 → h20R} /. ruleLMD;
```

Define new formulas for $k_n$and $h_n$, assuming that $\rho$0=$\rho$bar (fluid-crust density=density of the physical crust):

```
Clear[knBis, hnBis];
knBis[μhat_, ν_, χ_, ϵ_, ξ_, hn0_, n_] :=
 hn0 * (1 + (3 ξ)/(2 n + 1) * (hn0 * Λ[μhat, ν, ϵ, n]/(1 + Λ[μhat, ν, ϵ, n])))^-1 - 1
hnBis[μhat_, ν_, χ_, δρhat_, ϵ_, ξ_, hn0_, n_] := hn0 *
  (1 + (1 + (3 ξ)/(2 n + 1) * hn0) * Λ[μhat, ν, ϵ, n] + Λχ[ν, χ, ϵ, n] + Hρ[δρhat, 0, ν, ϵ, ξ, hn0, n])^-1
```

Now, define $k_n$and $h_n$ functions for the three density models, assuming that $\rho$0=$\rho$bar (the mantle is infinitely rigid):

```
h20RBis = hRLL0[1 - ϵ, ρ/ρb, ρbar/ρb, n] /. ruleTitan;
{knLRBis[ϵ_], knMRBis[ϵ_], knDRBis[ϵ_]} =
  knBis[μ/(ρ * g * R), ν, χ, ϵ, ρ/ρb, hn0, n] /. ruleTitan /. ruleElastic /.
    {hn0 → h20RBis} /. ruleLMD;
{hnLRBis[ϵ_], hnMRBis[ϵ_], hnDRBis[ϵ_]} =
  hnBis[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb, hn0, n] /. ruleTitan /. ruleElastic /.
    {hn0 → h20RBis} /. ruleLMD;
```

Plot the Love numbers as functions of crust thickness, for the two different fluid-crust densities:




**Conclusion:**
The results obtained with the different fluid-crust densities are equivalent to order $\epsilon$.
Comparing this figures with Figure 5 of Paper MW shows that the value of $k_2$ computed with

$\rho$0=$\rho$bar (dashed curves) is closer to the benchmark, especially for the density model 'Mixed' (red curve). This was expected since the $k_2$ formula, if $\rho$0=$\rho$bar, does not include any density corrections: the density structure of the whole body, crust included, is taken into account into the fluid-crust Love number.

## Example 4.5: Load Love numbers of Titan in function of crust thickness for different crust/ocean density models (not in MW)

For safety, clear variables to be replaced:

```
Clear[Rm, R, g, ϵ, hn0, n]
Clear[ρ, ρb, δρ, δρhat]
Clear[μ, ν, χ, χμ, μEm]
```

Replacement rules (the crust is supposed to be elastic) and fluid-crust Love number (it is a function of ($\rho$,$\delta\rho$,$\epsilon$)):

```
h20 = hVL0[Rm/R, ρ/ρb0[ρb, δρ, ϵ], μEm/(ρb0[ρb, δρ, ϵ] * g0[g, ρb, δρ, ϵ] * R), n] /. ruleTitan;
```

Gravity and radial load Love numbers (degree 2) of Titan in function of crust thickness;
three density models (Light, Mixed, Dense):

```
Clear[knLoadL, knLoadM, knLoadD, hnLoadL, hnLoadM, hnLoadD]
{knLoadL[ϵ_], knLoadM[ϵ_], knLoadD[ϵ_]} =
  knLoad[μ/(ρ * g * R), ν, χ, δρhat, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. ruleTitan /. ruleElastic /.
    {hn0 → h20} /. ruleLMD;
{hnLoadL[ϵ_], hnLoadM[ϵ_], hnLoadD[ϵ_]} =
  hnLoad[μ/(ρ * g * R), ν, χ, (ρ * g * R) * χμ, δρhat, ϵ, ρ/ρb, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /.
      ruleTitan /. ruleElastic /. {χμ -> χEice / μEice} /. {hn0 → h20} /. ruleLMD;
```

Plot as in Example 4.3:




## Example 4.6: Tilt factor of Europa as a function of crust thickness (MW-Fig 8)

For safety, clear variables to be replaced:

```
Clear[Rm, R, g, ϵ, hn0, n]
Clear[ρ, ρb, δρ, δρhat]
Clear[μ, ν, χ, μEm]
```

First, evaluate partially the fluid-crust Love number h20 with the replacement rule 'ruleEuropa';

don’t forget to use the fluid-crust bulk density and gravity;
h20 now depends on the three variables ($\rho$, $\delta\rho$, $\epsilon$):

```
h20 = hVL0[Rm/R, ρ/ρb0[ρb, δρ, ϵ], μEm/(ρb0[ρb, δρ, ϵ] * g0[g, ρb, δρ, ϵ] * R), n] /. ruleEuropa;
```

Next, evaluate partially the radial Love number of three rheology models (not strictly necessary for the tilt factor, but it is a better approximation as explained in Section 6.3 of MW).
The results (h2E,h2C,h2F) depend on the three variables ($\rho$, $\delta\rho$, $\epsilon$):

```
h2E = hn[μ/(ρ * g * R), ν, χ, δρ/ρ, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. {hn0 → h20} /. ruleEuropa /.
   ruleEbar;
h2C = hn[μ/(ρ * g * R), ν, χ, δρ/ρ, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. {hn0 → h20} /. ruleEuropa /.
   ruleCbar;
h2F = hn[μ/(ρ * g * R), ν, χ, δρ/ρ, ϵ, ρ/ρb0[ρb, δρ, ϵ], hn0, n] /. {hn0 → h20} /. ruleEuropa /.
   ruleFbar;
```

Then, express the tilt factor (degree 2) of Europa as a function of crust thickness;
Do it for the three rheology models (Elastic-like, Critical, Fluid-like) and for the three density models (Light, Mixed, Dense):

```
Clear[tiltEL, tiltEM, tiltED, tiltCL, tiltCM, tiltCD, tiltFL, tiltFM, tiltFD]
{tiltEL[ϵ_], tiltEM[ϵ_], tiltED[ϵ_]} =
  Abs[tilt[μ/(ρ * g * R), ν, χ, δρhat, 0, ϵ, h2E, n] /. ruleEuropa /. ruleLMD /. ruleEbar];
{tiltCL[ϵ_], tiltCM[ϵ_], tiltCD[ϵ_]} =
  Abs[tilt[μ/(ρ * g * R), ν, χ, δρhat, 0, ϵ, h2C, n] /. ruleEuropa /. ruleLMD /. ruleCbar];
{tiltFL[ϵ_], tiltFM[ϵ_], tiltFD[ϵ_]} =
  Abs[tilt[μ/(ρ * g * R), ν, χ, δρhat, 0, ϵ, h2F, n] /. ruleEuropa /. ruleLMD /. ruleFbar];
```

Finally, plot the tilt factor as in MW-Fig 8 (Light/Mixed/Dense = black/red/blue):




# 5. Dynamical Love numbers

## ▪ Example 5.1: Dynamical tilt factor of Europa as a function of ocean thickness (MW-Fig 14)

For safety, clear variables to be replaced:

```
Clear[Rm, R, g, n, qω, z, ϵ, ϵO]
Clear[ρ, ρb, δρhat]
Clear[μ, ν, χ, μEm]
```

First, evaluate partially the fluid-crust Love number h20 with the replacement rule ‘ruleEuropa’; It is sufficient here to approximate it by its value at zero crust thickness; the fluid-crust bulk density and gravity are thus the same as in the physical model.
h20 now depends on the variable $\rho$:

```
h20 = hVL0[Rm/R, ρ/ρb, μEm/(ρb * g * R), n] /. ruleEuropa;
```

The dynamical correction $\Lambda\omega$ depends on z, which is defined by $z = R_m / R_\epsilon$ and can be written in terms of $\epsilon$ and $\epsilon_O = d_O / R$, where $d_O$ is the ocean thickness:

```
zϵ = FullSimplify[Rm/Rϵ /. {Rm → R - d - d0, Rϵ → R - d} /. {d → ϵ * R, d0 → ϵ0 * R}]
```

$1 + \frac{\epsilon 0}{-1 + \epsilon}$

Express the dynamical tilt factor as a function of the relative crust thickness and of the relative ocean thickness.
Since the crust is elastic, use the replacement rule ‘ruleElastic’.
For the Light density model, use the replacement rule ‘ruleL’.

```
Clear[tiltDyn];
tiltDyn[ϵ_, ϵ0_] =
  (tilt[μ/(ρ * g * R), ν, χ, δρhat, Λωn[qω, z, n], ϵ, h20, n] * 1/ϵ) /. ruleElastic /. ruleEuropa /.
    {qω → qωEuropa, z -> zϵ} /. ruleL;
```

Assume that the crust thickness is either d=10 km or d=50 km:

```
{ϵ10, ϵ50} = {10., 50.} * 10^3 / RadiusEuropa;
```

Plot the dynamical tilt factor of Europa as a function of ocean thickness in units of d/R (MW-Fig 14):




## ■ Example 5.2: Dynamical resonance (MW-Fig 15)

For safety, clear variables to be replaced:

```
Clear[ξ0, δρhat, ϵ, qω, z, n, ρ, g, R]
Clear[μE, νE, χE, f, δ1, δ2]
```

The variable z can be replaced by zϵ as in Example 5.1:

```
zϵ = 1 + ϵ0/(-1 + ϵ);
```

Crust and ocean have the same density (1000 kg/m3) so that $\rho$=$\rho$Water and $\delta\rho$hat=0.
The crust is incompressible so that $\nu_E$=0.5 and $\chi_E$ = 0.
For the critical model, use the replacement rule ‘ruleC’.

```
Clear[k2Dyn, h2Dyn];
k2Dyn[ϵ_, ϵ0_] =
  knrDyn[μbar[μE, fϵ, δ1, δ2] / (ρ * g * R), νE, χE, δρhat, ϵ, qω, z, ξ0, n] /. {ξ0 → ρ / ρb, νE → 0.5, χE →
       0, δρhat → 0} /. {ρ → ρWater} /. ruleC /. ruleEuropa /. {qω → qωEuropa, z → zϵ};
h2Dyn[ϵ_, ϵ0_] = hnrDyn[μbar[μE, fϵ, δ1, δ2] / (ρ * g * R), νE, χE, δρhat, ϵ, qω, z, ξ0, n] /.
      {ξ0 → ρ / ρb, νE → 0.5, χE → 0, δρhat → 0} /. {ρ → ρWater} /.
    ruleC /. ruleEuropa /. {qω → qωEuropa, z -> zϵ};
```

Assume that the crust thickness is 10 km:

```
ϵ10 = 10. * 10^3 / RadiusEuropa;
```

Plot the dynamical resonance enhancing the Love numbers of Europa as a function of ocean thickness (MW-Fig 15);
assume that the crust thickness is 10 km:




Evaluate the resonant thickness (in meters) for d=10 km and d=50 km:

```
Clear[μE, fϵ, δ1, δ2, ϵ, n, ρ, g, R]
{ϵ10, ϵ50} = {10.*^3, 50.*^3} / RadiusEuropa;
Dresonant[R, qω, ξ, μbar[μE, fϵ, δ1, δ2] / (ρ * g * R), νbar[νE, fϵ, δ1, δ2], ϵ, n] /. {ξ → ρ / ρb} /.
    {ρ → ρWater} /. ruleEuropa /. {qω → qωEuropa} /. ruleE /. {{ϵ → ϵ10}, {ϵ → ϵ50}}
{157.437, 142.409}
```

# 6. A simple thick shell model: the homogeneous crust model (static and dynamical)

*This model is useful to estimate deviations from the membrane formulas due to the finite thickness of the crust.*
*In this model, the* $l_2 - h_2$ *and* $k_2 - h_2$ *relations depend on the factors* $z_h$ *and* $z_l$*:*

$k_2 + 1 = (1 + z_h \hat{\mu})\, h_2$

$l_2 = z_l\, h_2$

*The factors* $z_h$ *and* $z_l$ *depend on* $\hat{\mu}$ *and* $\Lambda_\omega$*, and on five geometrical factors* $(X_a, X_b, X_c, X_d, X_e)$*.*

## ■ Example 6.1: Dependence of geometrical factors on crust thickness (MP-Fig 13; MW-Table 11)

The geometrical factors are given by the following functions of x (x=(R-d)/R):

```
Xlist = {Xa[x], Xb[x], Xc[x], Xd[x], Xe[x]}
```

$$\left\{\frac{24\left(19-75x^3+112x^5-75x^7+19x^{10}\right)}{5\left(24+40x^3-45x^7-19x^{10}\right)},\ \frac{x^2\left(19+45x^3-40x^7-24x^{10}\right)}{24+40x^3-45x^7-19x^{10}},\ \frac{36-100x^3+308x^5-225x^7-19x^{10}}{5\left(24+40x^3-45x^7-19x^{10}\right)},\ \frac{x^2\left(3+5x^3-10x^7+2x^{10}\right)}{2\left(24+40x^3-45x^7-19x^{10}\right)},\ \frac{5x^2\left(1-x^3-x^7+x^{10}\right)}{24+40x^3-45x^7-19x^{10}}\right\}$$

The numerators and denominators can be factorized but the resulting expressions are more complicated:

```
Factor[Xlist]
```

$$\Big\{-\left(24(-1+x)\left(19+38x+57x^2+x^3-55x^4+x^5+57x^6+38x^7+19x^8\right)\right)/$$
$$\left(5\left(24+24x+24x^2+64x^3+64x^4+64x^5+64x^6+19x^7+19x^8+19x^9\right)\right),$$
$$\left(x^2\left(19+19x+19x^2+64x^3+64x^4+64x^5+64x^6+24x^7+24x^8+24x^9\right)\right)/$$
$$\left(24+24x+24x^2+64x^3+64x^4+64x^5+64x^6+19x^7+19x^8+19x^9\right),$$
$$\left(36+36x+36x^2-64x^3-64x^4+244x^5+244x^6+19x^7+19x^8+19x^9\right)/$$
$$\left(5\left(24+24x+24x^2+64x^3+64x^4+64x^5+64x^6+19x^7+19x^8+19x^9\right)\right),$$
$$-\left(x^2\left(-3-3x-3x^2-8x^3-8x^4-8x^5-8x^6+2x^7+2x^8+2x^9\right)\right)/$$
$$\left(2\left(24+24x+24x^2+64x^3+64x^4+64x^5+64x^6+19x^7+19x^8+19x^9\right)\right),$$
$$-\frac{5(-1+x)x^2\left(1+x+x^2\right)\left(1+x+x^2+x^3+x^4+x^5+x^6\right)}{24+24x+24x^2+64x^3+64x^4+64x^5+64x^6+19x^7+19x^8+19x^9}\Big\}$$

In the thin shell limit, the geometrical factors become (MW-Table 11):

```
Series[Xlist /. {x -> 1 - ϵ}, {ϵ, 0, 2}]
```

$$\left\{\frac{24\epsilon}{11}+\frac{96\epsilon^2}{121}+O[\epsilon]^3,\ 1-\frac{25\epsilon}{11}+\frac{175\epsilon^2}{121}+O[\epsilon]^3,\ \frac{3}{11}-\frac{32\epsilon}{121}-\frac{172\epsilon^2}{1331}+O[\epsilon]^3,\ \frac{1}{22}-\frac{7\epsilon}{242}-\frac{325\epsilon^2}{2662}+O[\epsilon]^3,\ \frac{3\epsilon}{11}-\frac{54\epsilon^2}{121}+O[\epsilon]^3\right\}$$

In the static limit, the membrane approximations of $z_h$ and $z_l$ are, at leading order,

```
zhLin[x_] := 24/11 (1 - x);
zlMemb = 3/11;
```

In the homogeneous body limit, the geometrical factors become (MW-Table 11):

```
Xlist /. {x → 0}
```

$$\left\{\frac{19}{5},\ 0,\ \frac{3}{10},\ 0,\ 0\right\}$$

Plot the factors $z_h$ and $z_l$ if $\Lambda\omega$=0;
plot also the membrane approximations given by MP-115 (MP-Fig 13A):







Zoom on the above plots, in the range 0.8 < x < 1:

$z_h$ (solid), membrane approx. (dotted)




$z_l$ (solid), membrane approx. (dot)